

Implicit Differentiation

Douglas Meade, Ronda Sanders, and Xian Wu
Department of Mathematics

Overview

This lab provides experience working with functions defined implicitly.

Maple Essentials

- Important Maple commands introduced in this lab are:

Command	Description	Example
<code>display</code>	display plots in a single plot (need <code>plots</code> package)	<code>display([F,G],title='Fig1');</code>
<code>implicitplot</code>	create graph of function defined implicitly (need <code>plots</code> package)	<code>implicitplot(x*y=1,x=0..1,y=0..1);</code>
<code>pointplot</code>	plot points (need <code>plots</code> package)	<code>pointplot([1,2], color=red, symbolsize=18);</code>
<code>implicitdiff</code>	compute derivatives of functions defined implicitly	<code>implicitdiff(f,y,x);</code> <code>implicitdiff(f,y,x\$2);</code>
<code>fsolve</code>	compute a solution of equations numerically	<code>fsolve({f=1,g=x^2},{x,y});</code> <code>fsolve({f,g},{x,y},{x=0..1,y=0..2});</code>
<code>with</code>	load a Maple package	<code>with(plots): with(plots);</code>

- The *ImplicitDifferentiation* maplet is available from the course website:

<http://www.math.sc.edu/calclab/141L-S12/labs> → ImplicitDifferentiation

Related course material/Preparation

§3.5 of the Calculus Text and §4.4 of the Maple Text.

Assignment

Problem 32 of Calculus Text on Page 214 and your lab instructor will give other assignment for each section

Hint for using `implicitplot`: Start with a big range for both x and y in `implicitplot` to see the size of the view window the graph will display and then re-plot the graph with that view window for a better plot.

Activities

Problem 1: Find the equation of the tangent line to the curve $2(x^2 + y^2)^2 = 25(x^2 - y^2)$ at the point $(3, 1)$. Then graph the curve, the point, and the tangent line with a viewing window of $(-5,5) \times (-2,4)$.

Problem 2: Find all points where the tangent line to the graph of $x^2y - xy^2 = 2$ is horizontal or vertical. (Hint: The tangent line is vertical where $dx/dy = 0$.)

Problem 3: Find d^2y/d^2x and d^3y/d^3x if y is defined implicitly by $y + \sin y = x$.

Example Problem

- a) Use implicit differentiation to find dy/dx for the Folium of Descartes $x^3 + y^3 = 3xy$.
- b) Find the equation of the tangent line to the Folium of Descartes at the point $(3/2, 3/2)$.
- c) Graph the curve, the point, and the tangent with a viewing window of $(-3,3) \times (-4,3)$.
- d) At what point(s) in the first quadrant is the tangent line to the Folium of Descartes horizontal?

Steps:

1. Start a Maple session with `restart;` and load the Maple `plots` package. This package allows us to plot points, use the `display` command, use the commands for implicitly-defined functions, and more. Notice that we used `‘:’` instead of `‘;’`. The difference is that the maple does not display the output with `‘:’`.


```
> restart;
> with(plots):
```
2. For part a), simply assign the Folium of Descartes to, say, `FD`, then use command `implicitdiff` to find dy/dx .


```
> FD:=x^3 +y^3 =3*x*y;
> dydx:=implicitdiff(FD,y,x);
```

 (Notice that `implicitdiff(f,x,y);` computes dx/dy and `implicitdiff(f,y,x$n);` computes $d^n y/d^n x$. You will need them to do problem 2 and problem 3, respectively.)
3. Next, to find the tangent line, we need a point and a slope. The point $(3/2, 3/2)$ is given and we find the slope `m` by evaluating dy/dx at this point.


```
> m:= eval(dydx, {x=3/2, y=3/2});
```
4. Find the equation of the tangent line by the point-slope formula $y = m(x - x_1) + y_1$.


```
> L:=x-> m*(x-3/2)+3/2;
```
5. Next, write (and assign) commands to plot the curve, the point, and the tangent line. Write the commands separately using `‘:’` so Maple does not display the output yet. (In the first plot command, the option `numpoints=10000` will insure a smooth curve.)


```
> P1:= implicitplot(FD, x=-3..3, y=-4..3, numpoints=10000):
> P2:= pointplot([3/2,3/2], color=green, symbolsize=15):
> P3:= plot(L(x), x=-3..3, y=-4..3, color=blue, linestyle=DOT):
```
6. These plots can then be displayed on a single plot using the `display` command.


```
> display([P1, P2, P3], title='Figure 1');
```
7. From the graph, we can see that the answer to part d) is a point located approximately at $(1.2, 1.5)$. Since this point is on the curve and the $dy/dx = 0$ at this point, we can find it's location by solving those two equations.


```
> fsolve({FD,dydx=0},{x,y},{x=1..2,y=1..2});
```

 (For a numerical solution in a specified region, `fsolve` in general does a better job than `solve`.)

Additional Notes

The `ImplicitDifferentiation` maplet provides additional practice finding the slope of a curve at a point.